

A Jupiter-holdak tánca

Minden kezdő amatőr egyik legnagyobb élménye a Jupiter-holdak megpillantása. Tapasztalt amatőrök is gyakran gyönyörködnek az apró kis fénypontokban, amint lassan keringenek a Jupiter korongja körül. Ez a bolygó volt az, amelyet Galilei először pillantott meg távcsövén keresztül. Sokan rendszeresen észlelik a holdak jelenségeit. Tehát minden amatőr számára szükségesek az előrejelzések. Ilyen előrejelzéseket mindenki találhat az évkönyvekben és a különböző folyóiratokban. Ám miért ne számíthatnánk ki magunk a Jupiter-holdak helyzetét egy rövid program segítségével? Egy ilyen programot készítettem Commodore 64 számítógépre a Simon's Basic segédprogram segítségével. A program alapjául a Sky and Telescope 1989 júniusi száma és Jean Meeus Astronomical Formulae for Calculators című könyve szolgált.

A programban megpróbáltam elkerülni a speciális karaktereket, és törekedtem a minél egyszerűbb megoldásra. Miután betöltöttük a Simon's Basicet, a programot RUN paranccsal indíthatjuk. Vegyük példának Galilei első megfigyeléseit, amelyeket Sidereus Nuncis című munkájában közölt.

Galilei rajzai 1610-ben

január 7	*	*	○	*
január 8			○	* * *
január 10	*	*	○	
január 11	* *		○	
január 12		*	○	*
január 13	*		○	* *

A programnak az alapadatokat a következőképpen adjuk meg:

```

10 REM
20 REM JUPITER HOLDAK SZAMITASA
30 REM
40 GOTO 400
50 REM          HOLDAK RAJZOLASA
60 X=139+L*INT(X*3.5+.5)
70 PLOT X,V,1
80 IF X<143 AND X>135 THEN 100
90 PLOT X,167,T
100 V(I)=X
110 RETURN
120 REM          JUPITER RAJZOLASA
130 LINE 136,0,136,160,1
140 LINE 142,0,142,160,1
150 LINE 0,160,320,160,1
160 LINE 0,174,320,174,1
170 CIRCLE 139,167,4,4,1
180 PRINT 139,167,1
190 LINE 136,166,142,166,0
200 LINE 136,168,142,168,0
210 RETURN
220 REM          FELIRATOZAS
230 TEXT 0,175,L$,1,1,8
240 TEXT 272,175,R$,1,1,8
250 RETURN
260 Y$="EV "+STR$(Y)
270 M$="HONAP "+STR$(M)
280 TEXT 30,183,Y$,1,1,8
290 TEXT 173,183,M$,1,1,8
300 RETURN
310 TEXT 50,183,Y$,0,1,8
320 TEXT 173,183,M$,0,1,8
330 RETURN
340 REM          ADATOK
350 DATA 84.5506,203.405863,5.906
360 DATA 41.5015,101.2916323,9.397
370 DATA 109.9770,50.2345169,14.989
380 DATA 176.3586,21.4879802,26.364
390 REM          ----- FOPROGRAM -----
391 REM
400 PI=3.14159265:PRINT CHR$(147)
410 PRINT "MELYIK LEGYIK BALOLDALON"
420 INPUT "KELET VAGY NYUGAT (K/N) ";A$
430 IF A$="N" THEN 460
440 IF A$="K" THEN 480
450 GOTO 420
460 L=1:L$="NYUGAT":R$="KELET"
470 GOTO 490
480 L=-1:L$="KELET":R$="NYUGAT"
490 GOSUB 1030
500 N=J-2415021+F
510 P=PI/180:HIRE$=0
520 GOSUB 230
530 V=0:GOSUB 130
540 DL=INT((N-INT(N))*20+.5)/20
550 IF DL=.5 THEN GOSUB 910
560 MT=(358.476+.9856003*N)*PI
570 MJ=(225.328+.0030053*N)*PI
580 JJ=221.647+.90251*N
590 VT=1.92*SIN(MT)+.02*SIN(2*MT)
600 VJ=5.55*SIN(MJ)+.17*SIN(2*MJ)
610 K=(JJ+VT-VJ)*PI
620 DT=SQR(28.07-10.406*COS(K))
630 Z=SIN(K)/DT
640 IN=ATN(Z/SQR(1-Z^2))
650 IN=IN/PI

```


```

660 F0=(N-DT/173)
670 F1=IN-VJ
680 RESTORE
690 FOR I=1 TO 4
700 READ A0,B0,C0
710 IF V(I)<0 THEN T=0:X=V(I):GOSUB 80
720 U(I)=A0+B0*F0+F1
730 X(I)=C0*SIN(U(I))*F+P1
740 T1=X*X(I):GOSUB 60
750 NEXT I
760 V=V+1:N=N+.05
770 IF V>=160 THEN 790
780 GOTO 540
790 TEXT 90,191,"FOLYTASSAM (I/N)",1,1,8
800 GET AS:IF AS="" THEN 800
810 IF AS<"N" THEN 510
820 TEXT 90,191,"FOLYTASSAM (I/N)",0,1,8
830 TEXT 90,191,"MASIK DATUM (I/N)",1,1,8
840 WAIT 190,1:GET AS
850 IF AS<"N" THEN CSET 0:GOTO 400
860 CSET 0:PRINT CHR$(147)
870 END
900 REM DATUM VALTOZTATO
910 D$=STR$(D)
920 IF LEN(D$)=2 THEN D$=" "+D$
930 TEXT 250,V+8,D$,1,1,8
940 LINE 5,V,25,V,1
950 LINE 255,V,275,V,1
960 IF Y$<" " THEN GOSUB 310
970 GOSUB 260
980 J=J+1:F=0:GOSUB 1170
990 RETURN
1010 REM DATUM ---> J.D.
1030 INPUT "EV,HONAP,NAP : ";Y,M,D
1040 D=INT(D)
1050 G=1:IF Y<=1585 THEN G=0
1060 D1=D:F=D-D1-.5
1070 J=-INT(7*(INT((M+9)/12)+Y)/4)
1080 IF G=0 THEN 1120
1090 S=SGN(M-9):A=ABS(M-9)
1100 J1=INT(Y+S*INT(A/7))
1110 J1=-INT((INT(J1/100)+1)*3/4)
1120 J=J+INT(275*M/9)+D1+G*J1
1130 J=J+1721027+2*G+367*Y
1140 IF F>0 THEN 1160
1150 F=F+1:J=J-1
1160 J=J+1:RETURN
1170 REM J.D. ---> DATUM
1180 G=1:IF Y<=1585 THEN G=0
1190 F=F+.5:IF F<1 THEN 1210
1200 F=F-1:J=J+1
1210 IF G=1 THEN 1230
1220 A=J:GOTO1250
1230 A1=INT((J/36524.25)-51.12264)
1240 A=J+1+A1-INT(A1/4)
1250 B=A+1524
1260 C=INT((B/365.25)-.3343)
1270 D=INT(365.25*C)
1280 E=INT((B-D)/30.61)
1290 D=B-D-INT(30.61*E)+.5
1300 M=E-1:Y=C-4716
1310 IF E>13.5 THEN M=M-12
1320 IF M<2.5 THEN Y=Y+1
1330 D=D-.5:RETURN

```

MELYIK LEGYEN BALOLDALON
KELET VAGY NYUGAT (K/N) ? K
EV,HONAP,NAP : ? 1610,1,7

A dátumot mindig számokkal írjuk be, tehát a kezdő dátumban csak számok szerepelhetnek vesszővel elválasztva. Az adatok megadása után a program elsőként a polgári dátumból Julián-dátumot számít és a grafikus képernyőre lassan megrajzolja a holdak helyzetét. A kezdő dátumtól kezdődően folyamatosan kirajzolja a holdak helyzetét nyolc napon keresztül. A különböző évkönyvekben már megszokott szinuszgörbéhez hasonlító rajz alatt folyamatosan kijelzi a program a Jupitert és holdjainak állását, ahogy a távcsőben látszanának. Az ábrán is látható rajz elkészítése öt percig tart.



Ha most összevetjük ábránkat Galilei rajzaival és figyelembe vesszük, hogy ezek a megfigyelések az éjszaka első óráiban készültek, jó egyezést találunk.

Ha valakinek van nyomtatója, és ugyanilyen rajzot szeretne készíteni, a következő sort kell beszúrnia a programsorok közé:

791 COPY

Ez az utasítás a képernyő tartalmát három perc alatt a nyomtatóra másolja.

ZALEZSÁK TAMÁS